Celestial Navigation

with

Sextant and Calculator HP-42S



Ackermans Harry

2019

Abbreviations and symbols

$LOP1$ = Line of position 1

$LOP2$ = Line of position 2

$COP1$ = Circle of position 1

$COP2$ = Circle of position 2

$H1$ = Observed altitude (Ho) in $COP1$

$H2$ = Observed altitude (Ho) in $COP2$

Dec = Declination

GHA = Greenwich Hour Angle

$D1$ = Dec of the sun in $COP1$

$G1$ = GHA of the sun in $COP1$

$D2$ = Dec of the sun in $COP2$

$G2$ = GHA of the sun in $COP2$

φ = Latitude of the FIX in the calculations of the COP

L = Longitude of the FIX in the calculations of the COP

$\Delta\varphi$ = Displacement in latitude of the observer

ΔL = Displacement in longitude of the observer

lat_M = Mean latitude

$latx$ = Dead-reckoning or assumed observer's latitude

$lonx$ = Dead-reckoning or assumed observer's longitude

FIX = Calculated position

laf = FIX latitude

lof = FIX longitude

$IN1, IN2$ = Intercept of LOP1, LOP2

$AZ1, AZ2$ = Azimuth of LOP1, LOP2

Ha = Apparent altitude

Dog = Distance over Ground of the observer

Cog = Course over Ground of the observer

Ie = Index error of the sextant

$Corr$ = Correction of the sextant height according to its certificate

HoE = Height of eye of the observer

$Da.Mo$ = Day and Month of measurement

$Year$ = Year of measurement

$U.T$ = Universal Time of measurement

$ALTs$ = Measured sextant height

$Limb$ = Top edge, bottom edge or center line of the sun

$Npos$ = Northern intersection of the circles of equal altitude

$Spos$ = Southern intersection of the circles of equal altitude

$Tran$ = Noon latitude of the observer (Transit)

Foreword

The astronomical positioning by the simultaneous measurement of a number of stars leads to a fairly accurate position, but the exercise necessary to identify those stars and to determine their height above the horizon at twilight will often be absent. That is why we limit ourselves to determine our position with the help of the sun.
Using nautical almanacs such as the Macmillan and the HO249 tables, we can traditionally determine our position on the basis of sun's altitudes.
With the advent of the programmable calculators, the entire search, calculation and plot work that follows could be taken over from us. *GHA* and *Dec* of the sun can be calculated with a built-in algorithm, as well as the azimuth and intercept of the lines of position. The somewhat difficult chart work is also digitally reproduced with the final result our position in latitude and longitude *(FIX).*

HP-42S

The HP-42S RPN Scientific is a programmable RPN calculator launched by HP in 1988. Its production was discontinued in 1995 but it is still regarded today as one of the best ever made in terms of quality and ease of programming. Its popularity was so great that he is offered on the site of Thomas Okken under the name "**Free42**: An HP-42S Calculator Simulator" as freeware for many operating systems of *computers and smartphones*. The Free42 computer application is delivered as a zip file in decimal and binary version but it is recommended to install Free42Decimal.exe. Android and iOS applications are in decimal version.
The Free42 app has the advantage that the programs can be saved and loaded as files, backups can be made and loaded into other computers. Instructions can be found here. Moreover, the calculation speed is many times higher than with the hardware version of the HP-42S.
Free42 programs are saved with the file extension ".raw".
In the astronavigation programs described below, only a limited number of keys is used, so that no thorough knowledge of the calculator is required. Those who want to deepen further can go here.

Programs

This text describes two programs for Free42 and HP-42S to navigate by the sun by means of a sextant or theodolite:

ASTRO_TSO.raw: calculates the position *(FIX)* using the azimuth-intercept method of Marcq St. Hilaire. This method, developed in 1875, gradually became the standard procedure for astronomical positioning and is based on the choice of an assumed position (A.P), as close as possible to the true position or the estimated position, and the calculation of the intersection of two so-called *lines of position* (LOP further in the text). This program makes the use of plotting sheets superfluous.

ASTRA_TSO.raw: calculates the position *(FIX)* using an alternative method that directly calculates the intersections of two *circles of position* (COP further in the text). For a *stationary* observer, contrary to the previous method, no assumed position (A.P) is required so that one can determine his position without any idea of where one is. In the case of a *moving observer*, as in the previous program, the latitude of the observer, $latx$, must be known for the calculation of the displacement in longitude ΔL.
Both programs have a built-in almanac of the sun with a maximum deviation of ± 0.5' and an average deviation of ± 0.3' for *GHA* and *Dec*. For verification, the almanac data on the sites backbearing.com and siranah.de can be consulted.

Corrections to the measured sextant height $\boldsymbol{ALTs}$

A number of corrections are applied to the measured sextant height in both programs;

- *Instrument correction:* if the sextant is provided with a certificate, this correction can be taken into account for with $Corr$ (in seconds). If not, the input is = 0
- *Index correction:* is a correction for the index error Ie of the sextant which in principle has to be measured before each new observation. Ie is negative if "off the arc" and positive "on the arc".
- *Dip:* this is a correction that is function of the height of eye HoE. When using an artificial horizon $HoE = 0$.

$$Dip['] \approx 1.76 \cdot \sqrt{HoE[m]}$$

- *Refraction:* this is a correction for the refraction of light according to Bennett's formula.

$$Refraction['] = \frac{1}{\tan\left(Ha[°] + \frac{7.31}{Ha[°] + 4.4}\right)}$$

- *Parallax:* as the horizontal parallax of the sun is very small (maximum ± 0.15'), no correction for parallax is applied.
- *Semi diameter (S.D):* this correction is made if the bottom or top edge *(Limb)* of the sun is measured. The *S.D* is calculated from the mean anomaly *(AM)* of the sun with the formula

$$S.D[°] = \frac{0.267}{1 - 0.017\,.\cos AM}$$

Sextant heights

$Ha = ALTs \pm Ie - Dip$ (Apparent altitude)

$Ho = Ha - Refraction \pm S.D$ (Observed altitude)

$HoE = 0$ implies the use of an artificial horizon

When using an artificial horizon $Dip = 0$ and $Ha = (ALTs \pm Ie)/2$

Used formulas

- If the Cog and Dog are known then the observer's displacement in latitude and longitude are respectively:

$$\Delta\varphi[°] = Dog[nm] \,.\, \cos Cog \,/60$$

$$\Delta L[°] = \frac{Dog[nm] \,.\, \sin Cog}{\cos lat_M \,\,.\, 60}$$

in which the mean latitude

$$lat_M = latx + \Delta\varphi/2$$

and $latx$ is the (assumed) latitude before displacement.

- In *ASTRO_TSO.raw* (azimuth-intercept method) the *FIX*-coordinates are

$$laf = \frac{IN2 \,.\, \sin AZ1 - IN1 \,.\, \sin AZ2}{\sin(AZ1 - AZ2)} + latx$$

$$lof = \frac{IN1 \,.\, \cos AZ2 - IN2 \,.\, \cos AZ1}{\sin(AZ1 - AZ2) \,.\, \cos latx} + lonx$$

where $latx$ and $lonx$ are the coordinates of the A.P after displacement. These values are calculated from the A.P before displacement, Cog and Dog.

Format and units of the input data

latx, lonx		+ (N and E) – (S and W)
Ie	x.xx [minutes]	– OFF the arc + ON the arc
HoE	xx.xx [meter]	Artificial horizon HoE=0
Da.Mo	xx.xx [day.month]	Ex: 5.07 23.12
U.T	xx.xx xx [h.min sec]	Ex: 17.0533 6.0806
latx, lonx, ALTs	xxx.xx x [ddd°mm.m’]	Ex: –15.416 156.093
Limb	Lower=–1 Center=0 Upper=1	
Dog	xxx.x [nautical miles]	
Cog	xxx.x [degrees]	
Corr	xx [seconds]	Ex: –17 12

Program components

ASTRO_TSO.raw: consists of 3 subroutines; *LOP1, LOP2*, and *TRAN*

- *LOP1 (Line of position1):* here the assumed position (A.P), which is usually the dead-reckoning position, and the data of the first observation are entered. The subroutine calculates the intercept *IN1* and azimuth *AZ1* of the first line of position *LOP1* and stores them in their variables with the same name.

- *LOP2 (Line of position2):* on the basis of the *Dog* and *Cog* a new A.P is defined. Together with the data of the second observation, the *IN2* and *AZ2* of *LOP2* are calculated relative to this new A.P. The coordinates of the intersection of the "offset" *LOP1* and *LOP2*, being the *FIX*, are obtained with the above formulas. From a geometric point of view this actually amounts to the translation of *LOP1* in the direction (= *Cog*) of the new A.P over a distance = *Dog*. The coordinates of the *FIX* are stored as "laf"and "lof".
  Since the intercept method ignores the curvatures of the actual position lines, the obtained *FIX* is not our exact position but rather an improved position (compared to AP). The residual error remains tolerable as long as the radii of the circles of equal altitude are not too small and AP is not too far from our actual position. The geometric error inherent to the intercept method can be decreased by iteration, i.e., substituting the obtained *FIX* for AP and repeating the calculations with the same altitudes and GP's (see appendix). This will result in a more accurate position. If necessary, we can reiterate the procedure until the obtained position remains virtually constant (rarely needed).

- *TRAN (Transit):* this part calculates the noon latitude of our position. The input of the estimated longitude $lonx$ only results in a rough estimate of the time of transit because it is calculated on the basis of a fixed transit time of the Greenwich meridian (12h U.T). It is more accurate to start from the actual "Greenwich meridian transit time" and to add 4 minutes per degree of western length or subtract 4 minutes per degree of eastern length.
  "South" and "North" indicate the position of the sun relative to the observer.
  The noon latitude is stored as "laf".

Notes

- *LOP1* and *LOP2* must be entered chronologically

- no Dip correction may be applied when using a bubble sextant and/or a theodolite. To make a distinction with the use of an artificial horizon *(HoE = 0)* and to prevent *Ha* from being divided by 2, set ***HoE = 0.0001*** and the Dip correction is negligible.

ASTRA_TSO.raw : consists of 3 subroutines; *COP1, COP2,* and *TRAN*

- *COP1 (Circle of position1):* $latx$ is the estimated or known latitude of the first observation and will be used in *COP2* to calculate the displacement in longitude ΔL. For a stationary observation (*Dog = 0*) this value is irrelevant.
  After inputting the other measured values, this routine calculates the parameters that define the *Circle of position1*, i.e. *GHA (= G1), Dec (= D1)* and *Ho (= H1).*
- *COP2 (Circle of position2):* with Dog and Cog we calculate the displacement of the first *circle of position COP1*, being $\Delta\varphi$ in latitude and ΔL in longitude. With the other measured values, this routine calculates the parameters that define the *Circle of position2*, i.e. *GHA (= G2), Dec (= D2), Ho (= H2)*, and ultimately our position *(FIX).*

  These 2 circles normally have 2 intersections on the globe. These are our 2 possible positions. When calculating the coordinates of these points of intersection, the choice is made between the northern intersection point ($Npos$) and the southern intersection point ($Spos$). The coordinates of the *FIX* are stored as "laf" and "lof".
  The algorithm for calculating the intersections is discussed in detail in the appendix.

- *TRAN (Transit):* see *ASTRO_TSO.raw*

Notes: see *ASTRO_TSO.raw*

ASTRO* versus *ASTRA

- If one is completely in the dark about his position one can not directly enter a suitable A.P into ASTRO. In case of a *stationary measurement (Dog=0)*, ASTRA does not need this data. With 2 observations, preferably with a difference in azimuth between 30 ° and 150 °, an accurate position can be determined directly and apart from the normal measurement errors



- The iteration process of the ASTRA-program runs very fast in Free42 but can take a lot of time with the HP-42S under certain circumstances (H1≈H2). due to its low clock speed.

- ASTRA always iterates to the most accurate position. With ASTRO, at a great distance between the A.P and the true position and / or at large sextant heights, at least 1 iteration may be required with the calculated FIX as the new A.P.

Free42 and HP-42S commands

- Press the XEQ button. The list of programs that have been loaded appears in the display.

- Press the key below the name of the selected program and after viewing the label, press R/S. The main menu appears in the display, consisting of the 3 subroutines and the EXIT routine to close the program.



- Each routine is started with the key under its name. The requested data is entered via the number pad. Pay particular attention to the units and the format. For negative numbers, first enter the number followed by +/-. An incorrect input can be deleted with the ← key and with RTN you return to the main menu. Each input is followed by R/S.
In a selection menu, press the key below the desired choice. If the program stops with relevant information in the display, press R/S to continue.

- Entered data and calculated values are stored as variables and can be called up with RCL and scroll keys ▲ and ▼

- Calculated values *GHA, Dec, laf* and *lof* are stored in the format [ddd°mm.m']. *AZ1*, *IN1* and *S.D* in [°].

- The results of the observations can be printed out. The print function is activated by PRINT ▲ PON and disabled by PRINT ▲ POFF. The print is available for almost all operating systems via the File menu. With the iPhone the screen must be touched just below the status bar.

Appendix: calculation of the intersections of 2 circles of position

We apply the formulas of the nautical triangle to 2 observations;

$$\sin H1 = \sin D1 \,.\, \sin(\varphi - \Delta\varphi) + \cos D1 \,.\, \cos(\varphi - \Delta\varphi) \,.\, \cos(G1 + L - \Delta L) \qquad (1)$$

$$\sin H2 = \sin D2 \,.\, \sin\varphi + \cos D2 \,.\, \cos\varphi \,.\, \cos(G2 + L) \qquad (2)$$

(1) and (2) are the equations of 2 circles of position in which φ and L are the geographical coordinates of the position of the observer at the time of the second measurement, i.e. the *FIX*. $H1$ and $H2$ are the true sextant heights (after corrections) and $\Delta\varphi$ and ΔL are respectively the displacement in latitude and longitude after the first measurement.

The system of 2 equations with 2 unknowns (φ and L) can not be solved analytically but by iteration. There are 2 conditions attached to this process:
- the iteration must be started in the equation of the circle with the smallest sextant height H, i.e. the circle with the largest radius.
- since for 1 value of φ, 2 values of L are possible and vice versa the appropriate choice must be made here.

If *H2 ≤ H1*

then the starting value of φ (= D2) is placed in equation (2). This gives 2 possible L-values:
$L = Acos(\ldots) - G2$ and $L = -Acos(\ldots) - G2$. One chooses the most eastern (= positive) value and since Acos is always positive

$$L = Acos\left(\frac{\sin H2 - \sin D2 \,.\, \sin\varphi}{\cos D2 \,.\, \cos\varphi}\right) - G2$$

This L-value is placed in (1) and the new φ-value for (2) is calculated

$$\varphi = 2.Atan\left(\frac{\sin D1 \pm \sqrt{\sin^2 D1 - \sin^2 H1 + \cos^2 D1\ .\cos^2(G1 + L - \Delta L)}}{\sin H1 + \cos D1\,.\cos(G1 + L - \Delta L)}\right) + \Delta\varphi$$

with $+\sqrt{\ldots}$ for the northern intersection and $-\sqrt{\ldots}$ for the southern intersection.

If *H2 > H1*

then the starting value of φ (= D2) is placed in equation (1). This gives 2 possible L -values: $L = Acos(\ldots) - G1 + \Delta L$ and $L = -Acos(\ldots) - G1 + \Delta L$. One chooses the most western (= negative) value and since Acos is always positive

$$L = -Acos\left(\frac{\sin H1 - \sin D1\,.\,\sin(\varphi - \Delta\varphi)}{\cos D1\,.\cos(\varphi - \Delta\varphi)}\right) - (G1 - \Delta L)$$

This L-value is placed in (2) and the new φ-value for (1) is calculated

$$\varphi = 2.Atan\left(\frac{\sin D2 \pm \sqrt{\sin^2 D2 - \sin^2 H2 + \cos^2 D2\,.\cos^2(G2 + L)}}{\sin H2 + \cos D2\,.\cos(G2 + L)}\right)$$

with $+\sqrt{\ldots}$ for the northern intersection and $-\sqrt{\ldots}$ for the southern intersection.

In both cases the value of φ is calculated by substitution, by means of the so-called T-formulas, with variables $\tan((\varphi - \Delta\varphi)/2)$ and $\tan(\varphi/2)$ respectively.

The iterations are stopped when 2 consecutive values of φ differ by less than 0.001°

In the choices made in the iterative process described above, it is implicitly assumed that the geographic position (G.P) of the sun at the second observation lies to the west of the G.P of the first observation.

Appendix: iterations in ASTRO

A "running" fix is less accurate than a stationary fix. After all, the *Cog* and *Dog* can only be estimated because the effects of current and wind (drift) are not exactly known in most cases. For a stationary measurement *(Dog = 0)*, the found position can be refined by inputting the *FIX* as the new A.P with the following key sequence in the main menu:

LOP1 RCL LAF R/S RCL LOF R/S and repeating the calculations with the initial input parameters.

Interesting websites

http://thomasokken.com/free42/
https://www.celnav.de/page2.htm
http://www.teacupnavigation.net/CN.html
http://www.backbearing.com/almanac.html
http://www.siranah.de

List of equations used in the text

$$\sin H1 = \sin D1 \,.\, \sin(\varphi - \Delta\varphi) + \cos D1 \,.\, \cos(\varphi - \Delta\varphi) \,.\, \cos(G1 + L - \Delta L)$$

$$\sin H2 = \sin D2 \,.\, \sin\varphi + \cos D2 \,.\, \cos\varphi \,.\, \cos(G2 + L)$$

$$L = Acos\left(\frac{\sin H2 - \sin D2 \,.\, \sin\varphi}{\cos D2 \,.\, \cos\varphi}\right) - G2$$

$$\varphi = 2.Atan\left(\frac{\sin D1 \pm \sqrt{\sin^2 D1 - \sin^2 H1 + \cos^2 D1 \,.\, \cos^2(G1 + L - \Delta L)}}{\sin H1 + \cos D1 \,.\, \cos(G1 + L - \Delta L)}\right) + \Delta\varphi$$

$$L = -Acos\left(\frac{\sin H1 - \sin D1 \,.\, \sin(\varphi - \Delta\varphi)}{\cos D1 \,.\, \cos(\varphi - \Delta\varphi)}\right) - (G1 - \Delta L)$$

$$\varphi = 2.Atan\left(\frac{\sin D2 \pm \sqrt{\sin^2 D2 - \sin^2 H2 + \cos^2 D2 \,.\, \cos^2(G2 + L)}}{\sin H2 + \cos D2 \,.\, \cos(G2 + L)}\right)$$

$$\Delta\varphi[°] = Dog[nm] \,.\, \frac{\cos Cog}{60}$$

$$\Delta L[°] = \frac{Dog[nm] \,.\, \sin Cog}{\cos lat_M \,.\, 60}$$

$$lat_M = latx + \Delta\varphi/2$$

$$laf = \frac{IN2 \,.\, \sin AZ1 - IN1 \,.\, \sin AZ2}{\sin(AZ1 - AZ2)} + latx$$

$$lof = \frac{IN1 \,.\, \cos AZ2 - IN2 \,.\, \cos AZ1}{\sin(AZ1 - AZ2) \,.\, \cos latx} + lonx$$

$$Lon = \pm Acos\left(\frac{\sin Ho - \sin Dec \,.\, \sin Lat}{\cos Dec \,.\, \cos Lat}\right) - GHA$$

$$Dip['] \approx 1.76 \cdot \sqrt{HoE[m]}$$

$$Refraction['] = \frac{1}{\tan\left(Ha[°] + \frac{7.31}{Ha[°] + 4.4}\right)}$$

$$S.D[°] = \frac{0.267}{1 - 0.017 \,.\, \cos AM}$$

ASTRONAVIGATION - SUN HP 42S

```
001 LBL "ASTRO"

002 DEG
003 CF 02
004 CF 03
005 CF 29
006 "......□.ASTRO.□"
007 ├ "If.........zon"
008 AVIEW
009 STOP

010 LBL 00

011 "LOP1"
012 KEY 1 XEQ A
013 "LOP2"
014 KEY 2 XEQ B
015 "TRAN"
016 KEY 4 XEQ C
017 "EXIT"
018 KEY 6 GTO J
019 CLA
020 AVIEW
021 MENU
022 STOP
023 GTO 00

024 LBL A

025 XEQ 01
026 FC? 01
027 RTN
028 STO IN1   (°)
029 RCL 04
030 STO AZ1   (°)
031 XEQ D
032 RTN

033 LBL B

034 XEQ 10
035 FIX 01
036 INPUT Dog
037 INPUT Cog
038 COS
039 RCLx Dog
040 60
041 ÷
042 STO 23          =Δφ
043 RCL 18
044 +
045 XEQ 09
046 STO lat*
047 RCL Cog
048 SIN
049 RCLx Dog
050 RCL 18
051 RCL 23
052 2
053 ÷
054 +
055 COS
056 ÷
057 60
058 ÷
059 RCL 17
060 +
061 XEQ 09
062 STO lon*
063 XEQ 02
064 FC? 01
065 RTN
066 XEQ D
067 XEQ 03
068 SF 02
069 XEQ D
070 PRA
071 PROMPT
072 RTN

073 LBL C

074 XEQ 10
075 FIX 03
076 INPUT lon*
077 XEQ 08
078 15
079 ÷
080 12
081 X<>Y
082 -
083 >HMS
084 "Transit.."
085 PRA
086 ├ ".~."
087 AIP
088 ├ ":"
089 FP
090 100
091 x
092 AIP
093 ├ ".U.T"
094 PROMPT
095 XEQ 02
096 FC? 01
097 RTN

098 LBL c

099 RCL 01
100 90
101 RCL- 08
102 "SOUTH"
103 KEY 1 GTO 14
104 "NORTH"
105 KEY 4 GTO 13
106 CLA
107 PROMPT
108 GTO c

109 LBL 13

110 ±

111 LBL 14

112 +
113 XEQ 09
114 STO laf
115 SF 02
116 SF 03
117 XEQ D
118 CF 03
119 PRA
120 PROMPT
121 RTN

122 LBL 01

123 XEQ 10
124 FIX 03
125 INPUT lat*
126 XEQ 08
127 STO 18
128 INPUT lon*
129 XEQ 08
130 STO 17

131 LBL 02

132 FIX 02
133 INPUT Ie
134 INPUT HoE
135 XEQ I
136 INPUT Da.Mo
137 IP
138 STO 00          =JO
139 LAST X
140 FP
141 100
142 x
143 STO 15          =MO
144 RCL IND ST x
145 STO+ 00
146 FIX 00
147 INPUT Year
148 4
149 ÷
150 FP
151 STO 02          =AA
152 X=0?
153 XEQ a
154 FIX 04
155 INPUT U.T
156 >HR
157 24
158 ÷
159 STO+ 00         =J2
160 FIX 03
161 INPUT ALTs
162 FIX 00
163 INPUT Corr
164 INPUT Limb
165 RCL Year
166 2000
167 -
168 365.25
169 x
170 0.5
171 +
172 RCL- 02
173 365250
174 ÷
175 STO 14
176 RCL 00
177 LAST X
178 ÷
179 +
180 STO 01          =T
181 RAD
182 6283.01961
183 x
184 0.043179665
185 -
186 STO 02          =AM
187 XEQ 20
188 0.033417
189 RCL 02
190 SIN
191 x
192 RCL+ 02
193 RCL 02
194 2
```

```
195 x
196 SIN
197 0.0003489
198 x
199 +
200 RCL 01
201 0.300052641
202 x
203 4.938242632
204 +
205 +
206 STO 03
207 SIN
208 0.397777
209 x
210 ASIN
211 >DEG
212 STO 01        =DEC
213 XEQ 09
214 STO DEC
215 RCL 03
216 TAN
217 0.917482
218 x
219 ATAN
220 STO 02
221 RCL 03
222 COS
223 X<0?
224 XEQ b
225 RCL 00
226 360.98564735
227 x
228 RCL 14
229 360007.7
230 x
231 100.465
232 +
233 360
234 MOD           =TSO
235 +
236 360
237 ÷
238 FP
239 PI
240 x
241 2
242 x
243 RCL- 02
244 >DEG          =GHA
245 ENTER
246 XEQ 09
247 STO GHA
248 R↓
249 DEG
250 RCL lon*
251 XEQ 08
252 STO 05
253 +
254 STO 02        =LHA
255 RCL lat*
256 XEQ 08
257 STO 06
258 COS
259 RCL 01
260 SIN
261 x
262 RCL 06
263 SIN
264 RCL 01
265 COS
266 x
267 RCL 02
268 COS
269 x
270 -
271 RCL 01
272 COS
273 ±
274 RCL 02
275 SIN
276 x
277 X<>Y
278 >POL McMillan 159
279 X<>Y
280 X<0?
281 XEQ 12
282 STO 04        =AZ
283 RCL 01
284 SIN
285 RCL 06
286 SIN
287 x
288 RCL 01
289 COS
290 RCL 02
291 COS
292 x
293 RCL 06
294 COS
295 x
296 +
297 ASIN
298 STO 02        =Hc
299 RCL ALTs
300 XEQ 08
301 RCL Ie
302 60
303 ÷
304 -
305 RCL Corr
306 3600
307 ÷
308 +
309 RCL HoE
310 X=0?          Artif. Hor
311 GTO 16
312 SQRT
313 0.0293        =Dip
314 x
315 -

316 LBL 15

317 STO 03
318 4.4           =Refrac
319 +
320 7.31
321 X<>Y
322 ÷
323 RCL+  03
324 TAN
325 1/X
326 60
327 ÷
328 RCL 03
329 X<>Y
330 -
331 RCL Limb
332 RCL S.D
333 x
334 -
335 STO 08        =Ho
336 RCL- 02
337 STO 03        =INT
338 RTN

339 LBL 16

340 R↓
341 2
342 ÷
343 GTO 15

344 LBL a

345 1
346 STO 02
347 RCL 15
348 2
349 X>=Y?
350 RTN
351 1
352 STO+ 00
353 RTN
354 LBL b

355 PI
356 STO+ 02
357 RTN

358 LBL 03        =FIX

359 "If.......*.FIX.*"
360 AVIEW
361 RCL AZ1
362 SIN
363 RCLx 03
364 RCL 04
365 SIN
366 RCLx IN1
367 -
368 RCL AZ1
369 RCL- 04
370 SIN
371 STO 07
372 ÷
373 RCL+ 06
374 XEQ 09
375 STO laf
376 RCL 04
377 COS
378 RCLx IN1
379 RCL AZ1
380 COS
381 RCLx 03
382 -
383 RCL: 07
384 RCL 06
385 COS
386 ÷
387 RCL+ 05
388 ABS
389 180
390 X<Y?
391 GTO 11
392 LAST X

393 LBL 04

394 XEQ 09
395 STO lof
396 RTN

397 LBL D

398 CLMENU
399 CLA
400 RCL Da.Mo
401 AIP
402 ├"/"
```

```
FP
100
x
AIP
├ "/"
FIX 00
ARCL Year
├ ".□."
RCL U.T
AIP
├ ":"
FP
100
x
AIP
├ ":"
FP
100
x
AIP
FS? 02
GTO d
├ "IfAZ:."
ARCL 04
├ "°.INT:."
RCL 03
60
x
FIX 01
ARCL STx
PRA
PROMPT
RTN

LBL d

CF 02
├ "If"
RCL Iaf
ENTER
XEQ 07
R↓
X<0?
GTO 05
├ "N.."

LBL e

FS? 03
RTN
RCL Iof
ENTER
XEQ 07
R↓
X<0?
GTO 06
├ "E"
RTN

LBL 05

├ "S.."
GTO e

LBL 06

├ "W"
RTN

LBL 07

ABS
AIP
├ "°"
FP
100
x
FIX 01
RND
ARCL ST x
RTN

LBL 08        =>d.ddd

IP
LAST X
FP
5
x
3
÷
+
RTN

LBL 09  =>dd.mm.m

IP
LAST X
FP
3
x
5
÷
+
RTN

LBL 10

CLMENU
SF 01
"RTN"
KEY 6 GTO H
MENU
RTN

LBL 11

LAST X
ENTER
SIGN
360
x
-
GTO 04

LBL 12

360
+
RTN

LBL H

CF 01
RTN

LBL I

-1
STO 01
30
ST0 02
58
STO 03
89
STO 04
119
STO 05
150
STO 06
180
STO 07
211
STO 08
242
STO 09
272
STO 10
303
STO 11
333
STO 12
RTN

LBL 20        =>S.D

COS
0.017
x
1
X<>Y
-
0.267
X<>Y
÷
STO S.D
RTN

LBL J

CLMENU
EXITALL
CLST
DEG
FIX 04
END
```

Sight Reduction with 2 Lines of Position (with sextant or theodolite) and calculator HP 42S or App FREE42

lat*, lon* = + (N en E)	HP 42S 1347 bytes
- (S en W)	App Free42 1344 bytes
HoE = meter	Artificial Horizon HoE=0
Limb	lower= -1 centre= 0 upper= 1
Format lat*, lon*, ALTs	= xxx.xx x [ddd°.mm.m']
Format U.T	= xx.xx xx [h . min sec]
Format Da.Mo	= xx.xx [Day . Month]
Dog	= xxx.x [nautical miles]
Cog	= xxx.x [°]
Index error Ie	= - OFF the arc + ON the arc
Format Ie	= x.xx [minutes]
Format Corr	= xx [sec]

ASTRONAVIGATION - SUN HP 42S

```
001 LBL "ASTRA"

002 DEG
003 CF 02
004 CF 03
005 CF 29
006 "......□.ASTRA.□"
007 ├ "If.........zon"
008 AVIEW
009 STOP

010 LBL 00

011 "COP1"
012 KEY 1 XEQ A
013 "COP2"
014 KEY 2 XEQ B
015 "TRAN"
016 KEY 4 XEQ C
017 "EXIT"
018 KEY 6 GTO J
019 CLA
020 AVIEW
021 MENU
022 STOP
023 GTO 00

024 LBL A

025 XEQ 01
026 FC? 01
027 RTN
028 STO 20      =SIN H1
029 RCL 00
030 STO 21      =GHA1
031 RCL 01
032 STO 22      =DEC1
033 XEQ D
034 RTN

035 LBL B

036 XEQ 10
037 FIX 01
038 INPUT Dog
039 INPUT Cog
040 COS
041 RCLx Dog
042 60
043 ÷
044 STO 23      =Δφ
045 RCL Cog
046 SIN
047 RCLx Dog
048 RCL 19
049 RCL 23
050 2
051 ÷
052 +
053 COS         =Mean Lat
054 ÷
055 60
056 ÷
057 STO 24      =ΔL
058 XEQ 12
059 FC? 01
060 RTN
061 STO 02      =SIN H2
062 XEQ 02
063 FC? 01
064 RTN
065 RCL 11
066 XEQ 09
067 STO laf
068 RCL 10
069 ABS
070 180
071 X<Y?
072 GTO 11
073 LAST X

074 LBL b

075 XEQ 09
076 STO lof
077 SF 02
078 XEQ D
079 PRA
080 PROMPT
081 RTN

082 LBL C       =Transit

083 XEQ 10
084 FIX 03
085 INPUT lon*
086 XEQ 08
087 15
088 ÷
089 12
090 X<>Y
091 -
092 >HMS
093 "Transit.."
094 PRA
095 ├ ".~."
096 AIP
097 ├ ":"
098 FP
099 100
100 x
101 AIP
102 ├ ".U.T"
103 PROMPT
104 XEQ 12
105 FC? 01
106 RTN

107 LBL c

108 RCL 01
109 90
110 RCL- 08
111 "SOUTH"
112 KEY 1 GTO 16
113 "NORTH"
114 KEY 4 GTO 15
115 CLA
116 PROMPT
117 GTO c

118 LBL 15

119 ±

120 LBL 16

121 +
122 XEQ 09
123 STO laf
124 SF 02
125 SF 03
126 XEQ D
127 CF 03
128 PRA
129 PROMPT
130 RTN

131 LBL 01

132 XEQ 10
133 FIX 03
134 INPUT lat*  =DRP
135 XEQ 08
136 STO 19

137 LBL 12

138 FIX 02
139 INPUT Ie
140 INPUT HoE
141 XEQ I
142 INPUT Da.Mo
143 IP
144 STO 00      =JO
145 LAST X
146 FP
147 100
148 x
149 STO 15      =MO
150 RCL IND ST x
151 STO+ 00
152 FIX 00
153 INPUT Year
154 4
155 ÷
156 FP
157 STO 02      =AA
158 X=0?
159 XEQ 13
160 FIX 04
161 INPUT U.T
162 >HR
163 24
164 ÷
165 STO+ 00     =J2
166 FIX 03
167 INPUT ALTs
168 RCL Year
169 2000
170 -
171 365.25
172 x
173 0.5
174 +
175 RCL- 02
176 365250
177 ÷
178 STO 14
179 RCL 00
180 LAST X
181 ÷
182 +
183 STO 01      =T
184 RAD
185 6283.01961
186 x
187 0.043179665
188 -
189 STO 02      =AM
190 XEQ 20
191 0.033417
```

```
192  RCL 02
193  SIN
194  x
195  RCL+ 02
196  RCL 02
197  2
198  x
199  SIN
200  0.0003489
201  x
202  +
203  RCL 01
204  0.300052641
205  x
206  4.938242632
207  +
208  +
209  STO 03
210  SIN
211  0.397777
212  x
213  ASIN
214  >DEG
215  STO 01        =DEC
216  XEQ 09
217  STO DEC
218  RCL 03
219  TAN
220  0.917482
221  x
222  ATAN
223  STO 02
224  RCL 03
225  COS
226  X<0?
227  XEQ 14
228  RCL 00
229  360.98564735
230  x
231  RCL 14
232  360007.7
233  x
234  100.465
235  +
236  360
237  MOD           =TSO
238  +
239  360
240  ÷
241  FP
242  PI
243  x
244  2
245  x
246  RCL- 02
247  >DEG
248  STO 00        =GHA
249  XEQ 09
250  STO GHA
251  DEG
252  FIX 00
253  INPUT Corr
254  INPUT Limb
255  RCL ALTs
256  XEQ 08
257  RCL Ie
258  60            =Ie
259  ÷
260  -
261  RCL Corr
262  3600
263  ÷
264  +
265  RCL HoE
266  X=0?          Artif.Hor
267  GTO 18
268  SQRT
269  0.0293
270  x             =Dip
271  -

272  LBL 17

273  STO 03
274  4.4           =Refract
275  +
276  7.31
277  X<>Y
278  ÷
279  RCL+  03
280  TAN
281  1/X
282  60
283  ÷
284  RCL 03
285  X<>Y
286  -
287  RCL Limb
288  RCL S.D
289  x
290  -
291  STO 08        =Ho
292  SIN
293  RTN

294  LBL 13

295  1
296  STO 02
297  RCL 15
298  2
299  X>=Y?
300  RTN
301  1
302  STO+ 00
303  RTN

304  LBL 14

305  PI
306  STO+ 02
307  RTN

308  LBL 18

309  R↓
310  2
311  ÷
312  GTO 17

313  LBL 02

314  CF 04
315  "Npos"
316  KEY 1 GTO 22
317  "Spos"
318  KEY 4 GTO 21
319  CLA
320  PROMPT
321  GTO 02

322  LBL 21

323  SF 04

324  LBL 22

325  ".......*.FIX.*"
326  AVIEW
327  RCL 01
328  STO 05        =STARTφ
329  SIN
330  STO 06
331  LAST X
332  COS
333  STO 07        =COS D2
334  RCL 21        =G1
335  RCL- 24
336  STO 04        =G1- ΔL
337  RCL 22
338  SIN
339  STO 08        =SIN D1
340  LAST X
341  COS
342  STO 09        =COS D1
343  RCL 02
344  RCL 20
345  X<Y?
346  GTO 04

347  LBL 03

348  RCL 02        =SIN H2
349  RCL 06        =SIN D2
350  RCL 05
351  SIN
352  x
353  -
354  RCL÷ 07
355  RCL 05
356  COS
357  ÷
358  ACOS
359  RCL- 00
360  CPX?
361  GTO 04
362  STO 10        =L
363  RCL+ 04
364  COS
365  RCLx 09
366  STO 12
367  X↑2
368  RCL 20
369  X↑2
370  -
371  RCL 08
372  X↑2
373  +
374  X<0?
375  GTO 04
376  SQRT
377  FS? 04
378  ±
379  RCL+ 08
380  RCL 20
381  RCL+ 12
382  ÷
383  ATAN
384  2
385  x
386  RCL+ 23
387  STO 11        =φ
388  RCL- 05
389  ABS
390  0.001
391  X>=Y?
392  RTN
393  RCL11
394  STO 05
395  GTO 03

396  LBL 04

397  RCL 20        =SIN H1
```

```
RCL 08      =SIN D1
RCL 05
RCL- 23
STO 12
SIN
x
-
RCL 12
COS
÷
RCL÷ 09
ACOS
±
RCL- 04
CPX?
GTO 03
STO 10      =L
RCL+ 00     =G2+L
COS
RCLx 07
STO 12
X↑2
RCL 02
X↑2
-
RCL 06
X↑2
+
X<0?
GTO 03
SQRT
FS? 04
±
RCL+ 06
RCL 02
RCL+ 12
÷
ATAN
2
x
STO 11      =φ
RCL- 05
ABS
0.001
X>=Y?
RTN
RCL11
STO 05
GTO 04

LBL D

CLMENU
CLA
RCL Da.Mo
AIP
├"/"
FP
100
x
AIP
├"/"
FIX 00
ARCL Year
├".□."
RCL U.T
AIP
├":"
FP
100
x
AIP
├":"
FP
100
x
AIP
FS? 02
GTO d
├"IfALTs:."
FIX 03
ARCL ALTs
PRA
PROMPT
RTN

LBL d

CF 02
├"If"
RCL Iaf
ENTER
XEQ 07
R↓
X<0?
GTO 05
├"N.."

LBL e

FS? 03
RTN
RCL Iof
ENTER
XEQ 07
R↓
X<0?
GTO 06
├"E"
RTN

LBL 05

├"S.."
GTO e

LBL 06

├"W"
RTN

LBL 07

ABS
AIP
├"°"
FP
100
x
FIX 01
RND
ARCL ST x
RTN

LBL 08      =>d.ddd

IP
LAST X
FP
5
x
3
÷
+
RTN

LBL 09 =>dd.mm.m

IP
LAST X
FP
3
x
5
÷
+
RTN

LBL 10

CLMENU
SF 01
"RTN"
KEY 6 GTO H
MENU
RTN

LBL 11

LAST X
ENTER
SIGN
360
x
-
GTO b

LBL H

CF 01
RTN

LBL I

-1
STO 01
30
ST0 02
58
STO 03
89
STO 04
119
STO 05
150
STO 06
180
STO 07
211
STO 08
242
STO 09
272
STO 10
303
STO 11
333
STO 12
RTN

LBL 20      =>S.D

COS
0.017
x
1
X<>Y
-
0.267
X<>Y
÷
STO S.D
RTN
```

```
595  LBL J

596  CLMENU
597  EXITALL
598  CLST
599  DEG
600  FIX 04
601  END
```

Sight Reduction with 2 Circles of Position (with sextant or theodolite) and calculator HP 42S or App FREE42

lat*, lon* = + (N en E)	HP 42S 1386 bytes
- (S en W)	App Free42 1381 bytes
HoE = meter	Artificial Horizon HoE=0
Limb	lower= -1 centre= 0 upper= 1
Format lat*, lon*, ALTs	= xxx.xx x [ddd°.mm.m']
Format U.T	= xx.xx xx [h . min sec]
Format Da.Mo	= xx.xx [Day . Month]
Dog	= xxx.x [nautical miles]
Cog	= xxx.x [°]
Index error Ie	= - OFF the arc + ON the arc
Format Ie	= x.xx [minutes]
Format Corr	= xx [sec]